

```

//*****
//*** INSERT IF FIELD MACRO
//*** HOTDOCS DOCUMENT ASSEMBLER V3.0
//*** CAPSOFT DEVELOPMENT CORPORATION
//*****

revealOn := ?RevealCodesActive
//*** CHECK FOR EXISITING FIELD
RUN("hdchkfld.wcm")

// CHECK FOR END IF
blockActive := ?BlockActive
ifMode := 1
endIfStr := SUBSTR(?SelectedText;1;8)
IF (blockActive <> 0 AND endIfStr = "«END IF»")
    ifMode := 7
ENDIF

//*** CALL LOAD AND CALL THE DLL
Result := 0
SelStr := ""
OrgFile := ?Path+?Name
IF (?MajorVersion <= 6)
    DllLoad(HotdocsDll;"HD3.dll")
    wpid = 32 * 4
ELSE
    DllLoad(HotdocsDll;"HD3_32.dll")
    wpid = 32 * 8
ENDIF
Display(On!)
DllCall(
    HotdocsDll;
    "InsertIF";
    RetStr:AnsiString;
    {OrgFile;ifMode+wpid;ADDRESS(Result);SelStr}
)
Display(Off!)
DllFree(HotdocsDll)

//*** IF SOMETHING IS WRONG, QUIT
IF (Result=0)
    IF (revealOn)
        RevealCodes(On!)
    ENDIF
    Quit
ENDIF

```

```

/**** SET A BOOK MARK AT THE END OF SELECTED TEXT
/**** POSITION AT THE BEGINNING OF SELECTED TEXT
selBM := "CapsBM"
botBM := "CapsBotBM"
BookmarkDelete(selBM)
BookmarkDelete(botBM)
IF (blockActive <> 0)
    BookmarkCreate(selBM)
    BookmarkBlock(selBM)
    PosSelectBottom()
    SelectMode(Off!)
    BookmarkCreate(botBM)
    BookmarkBlock(selBM)
    BookmarkDelete(selBM)
    PosSelectTop()
ENDIF
SelectMode(Off!)
IF (ifMode = 7)
    FOR (i;0;i<8;i+1)
        DeleteCharNext()
    ENDFOR
ENDIF

/**** ADD IF CONDITION IN CHEVRONS FOLLOWED BY HARD RETURN
IF (qcOn) QuickCorrect(Off!) ENDIF
IF (cflOn) CapBeginSentence(Off!) ENDIF
IF (ssqOn) SingleSmartQuote(Off!;;) ENDIF
IF (dsqOn) DoubleSmartQuote(Off!;;) ENDIF
IF (ticOn) IrregularCap(Off!) ENDIF

CALL(InsField@)
IF (ifMode = 1)
    HardReturn()
ENDIF

/**** MOVE TO THE END OF THE SELECTED TEXT
IF (blockActive <> 0)
    BookmarkFind(botBM)
    BookmarkDelete(botBM)
ENDIF

/**** ADD ENDIF IN CHEVRONS FOLLOWED BY HARD RETURN
RetStr := "END IF"
CALL(InsField@)
HardReturn()
IF (qcOn) QuickCorrect(On!) ENDIF
IF (cflOn) CapBeginSentence(On!) ENDIF

```

```
IF (ssqOn) SingleSmartQuote(On!;;) ENDIF
IF (dsqOn) DoubleSmartQuote(On!;;) ENDIF
IF (ticOn) IrregularCap(On!) ENDIF
IF (revealOn) RevealCodes(On!) ENDIF
QUIT
```

```
LABEL(InsField@)
  Type("<<>")
  PosCharPrevious()
  Type("X")
  SelectCharPrevious
  TextColor(;1;2;254)
  PosSelectBottom()
  SelectMode(Off!)
  DeleteCharPrevious()
  Type(RetStr)
  PosCharNext()
RETURN
```